```
> restart; with(linalg):
```

> # **Problem 1**

```
> L:=u->diff(u,x,x)+diff(u,y,y);
  G:=u->diff(u,x)^2+diff(u,y)^2;
  N:=u->(r-1)*u-2*L(u)-L(L(u))+diff(G(u)*diff(u,x),x)+diff(G(u)*
  diff(u,y),y);
```

$$L := u \to \frac{\partial^2}{\partial x^2} u + \frac{\partial^2}{\partial y^2} u$$

$$G := u \to \left(\frac{\partial}{\partial x} u\right)^2 + \left(\frac{\partial}{\partial y} u\right)^2$$

$$N := u \to (r-1)\, u - 2\, L(u) - L(L(u)) + \frac{\partial}{\partial x}\left(G(u)\left(\frac{\partial}{\partial x} u\right)\right) + \frac{\partial}{\partial y}\left(G(u)\left(\frac{\partial}{\partial y} u\right)\right) \quad \textbf{(1)}$$

> # (a) Growth rate:

```
> solve(factor(convert(series(combine(N(a*cos(q*x)),trig),a,2),
  polynom))=sigma*a*cos(q*x),{sigma});
```

$$\{\sigma = r - 1 + 2\, q^2 - q^4\} \quad \textbf{(2)}$$

> # (b) Amplitude of saturated stripe state

```
> eq:=combine(N(AS*cos(x)),trig):
  sls:=solve(coeff(eq,cos(x),1),{AS});
```

$$sls := \{AS = 0\}, \left\{AS = \frac{2}{3}\sqrt{3}\,\sqrt{r}\right\}, \left\{AS = -\frac{2}{3}\sqrt{3}\,\sqrt{r}\right\} \quad \textbf{(3)}$$

> # (c) Amplitude of saturated square state

```
> eq:=combine(N(AL*(cos(x)+cos(y))),trig):
  cf:=coeff(eq,cos(x),1);
  sll:=solve(cf,{AL});
```

$$cf := AL\, r - \frac{5}{4} AL^3$$

$$sll := \{AL = 0\}, \left\{AL = \frac{2}{5}\sqrt{5}\,\sqrt{r}\right\}, \left\{AL = -\frac{2}{5}\sqrt{5}\,\sqrt{r}\right\} \quad \textbf{(4)}$$

> # **Problem 2**

> # Stability of square state

```
> u1:=subs(sll[2],AL*(cos(x)+cos(y))+b*cos(x)+c*cos(y));
  eq:=combine(N(u1),trig):
```

$$u1 := \frac{2}{5}\sqrt{5}\,\sqrt{r}\,(\cos(x) + \cos(y)) + b\cos(x) + c\cos(y)$$

```
> dbdt:=coeff(eq,cos(x),1);
  dcdt:=coeff(eq,cos(y),1);
  A:=subs({b=0,c=0},matrix(2,2,[diff(dbdt,b),diff(dbdt,c),diff
  (dcdt,b),diff(dcdt,c)]));
  e:=eigenvectors(A);
```

$$dbdt := -\frac{2}{5}\sqrt{5}\sqrt{r}\,c\,b - \frac{3}{4}b^3 - \frac{6}{5}r\,b - \frac{4}{5}r\,c - \frac{1}{2}c^2 b - \frac{9}{10}\sqrt{5}\sqrt{r}\,b^2 - \frac{1}{5}c^2\sqrt{5}\sqrt{r}$$

$$dcdt := -\frac{2}{5}\sqrt{5}\sqrt{r}\,c\,b - \frac{3}{4}c^3 - \frac{1}{2}b^2 c - \frac{4}{5}r\,b - \frac{6}{5}r\,c - \frac{9}{10}c^2\sqrt{5}\sqrt{r} - \frac{1}{5}\sqrt{5}\sqrt{r}\,b^2$$

$$A := \begin{bmatrix} -\frac{6}{5}r & -\frac{4}{5}r \\ -\frac{4}{5}r & -\frac{6}{5}r \end{bmatrix}$$

$$e := \left[-2\,r, 1, \left\{\begin{bmatrix} 1 & 1 \end{bmatrix}\right\}\right], \left[-\frac{2}{5}r, 1, \left\{\begin{bmatrix} -1 & 1 \end{bmatrix}\right\}\right] \tag{5}$$

> # Both eigenvalues $-2r$ and $-\frac{2}{5}r$ are negative for $r > 0$, so squares are stable

> # Stability of stripe state

```
> u1:=subs(sls[2],AS*cos(x)+b*cos(x)+c*cos(y));
  eq:=combine(N(u1),trig):
```

$$u1 := \frac{2}{3}\sqrt{3}\sqrt{r}\cos(x) + b\cos(x) + c\cos(y)$$

```
> dbdt:=coeff(eq,cos(x),1);
  dcdt:=coeff(eq,cos(y),1);
  A:=subs({b=0,c=0},matrix(2,2,[diff(dbdt,b),diff(dbdt,c),diff
  (dcdt,b),diff(dcdt,c)]));
  e:=eigenvectors(A);
```

$$dbdt := -2\,r\,b - \frac{3}{4}b^3 - \frac{1}{2}c^2 b - \frac{3}{2}\sqrt{3}\sqrt{r}\,b^2 - \frac{1}{3}c^2\sqrt{3}\sqrt{r}$$

$$dcdt := \frac{1}{3}r\,c - \frac{3}{4}c^3 - \frac{1}{2}b^2 c - \frac{2}{3}c\sqrt{3}\sqrt{r}\,b$$

$$A := \begin{bmatrix} -2\,r & 0 \\ 0 & \frac{1}{3}r \end{bmatrix}$$

$$e := \left[\frac{1}{3}r, 1, \left\{\begin{bmatrix} 0 & 1 \end{bmatrix}\right\}\right], \left[-2\,r, 1, \left\{\begin{bmatrix} 1 & 0 \end{bmatrix}\right\}\right] \tag{6}$$

> # One of the eigenvalues, $\frac{1}{3}r$, is positive for $r > 0$, so stripes are unstable

> # **Problem 3**

```
> restart;
  with(plots):
  with(linalg):
  interface(showassumed=0);
  assume(r>0); additionally(r>(1-q^2)^2):
  _EnvExplicit:=true:
```

$$1$$

```
> L:=u->diff(u,x,x)+diff(u,y,y);
  N:=u->(r-1)*u-2*L(u)-L(L(u))-u^3;
```

$$L := u \rightarrow \frac{\partial^2}{\partial x^2}\, u + \frac{\partial^2}{\partial y^2}\, u$$

$$N := u \rightarrow (r-1)\, u - 2\, L(u) - L(L(u)) - u^3 \tag{7}$$

```
> eq:=collect(combine(N(a1*cos(x))),{cos(x)});
  cf:=factor(coeff(eq,cos(x),1));
  s1:=solve(cf,a1);
```

$$eq := \left(a1\, r\sim - \frac{3}{4}\, a1^3 \right) \cos(x) - \frac{1}{4}\, a1^3 \cos(3\, x)$$

$$cf := -\frac{1}{4}\, a1\, (-4\, r\sim + 3\, a1^2)$$

$$s1 := 0,\ \frac{2}{3}\, \sqrt{3}\, \sqrt{r\sim},\ -\frac{2}{3}\, \sqrt{3}\, \sqrt{r\sim}$$

```
> # Zig-zag instability:
> eq:=collect(simplify(subs(a=sqrt(4/3*(r-(1-q^2)^2)),combine
  (coeff(N(a*cos(q*x)+epsilon*(alpha*cos(q*x)+beta*sin(q*x))*exp
  (I*Q*y)),epsilon,1),trig))),{cos(q*x),sin(q*x)});
```

$$eq := -e^{I\,Q\,y} (2\, r\sim\, \alpha + Q^4\, \alpha + 2\, Q^2\, \alpha\, q\sim^2 - 2\, Q^2\, \alpha - 2\, \alpha - 2\, \alpha\, q\sim^4 + 4\, \alpha\, q\sim^2) \cos(q\sim x) - e^{I\,Q\,y} (2\, Q^2\, \beta\, q\sim^2 - 2\, Q^2\, \beta + Q^4\, \beta) \sin(q\sim x) - e^{I\,Q\,y} (-\alpha \cos(3\, q\sim x)\, q\sim^4 - \beta \sin(3\, q\sim x) + \beta \sin(3\, q\sim x)\, r\sim + 2\, \beta \sin(3\, q\sim x)\, q\sim^2 - \beta \sin(3\, q\sim x)\, q\sim^4 - \alpha \cos(3\, q\sim x) + 2\, \alpha \cos(3\, q\sim x)\, q\sim^2 + \alpha \cos(3\, q\sim x)\, r\sim)$$

```
> A11:=simplify(coeff(coeff(eq*exp(-I*Q*y),cos(q*x),1),alpha,1)):
  A12:=simplify(coeff(coeff(eq*exp(-I*Q*y),cos(q*x),1),beta,1)):
  A21:=simplify(coeff(coeff(eq*exp(-I*Q*y),sin(q*x),1),alpha,1)):
  A22:=simplify(coeff(coeff(eq*exp(-I*Q*y),sin(q*x),1),beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvalues(A);
```

$$A := \begin{bmatrix} -2\, r\sim - Q^4 - 2\, Q^2\, q\sim^2 + 2\, Q^2 + 2 + 2\, q\sim^4 - 4\, q\sim^2 & 0 \\ 0 & -Q^2\, (2\, q\sim^2 - 2 + Q^2) \end{bmatrix}$$

$$e1 := -2\,r\sim - Q^4 - 2\,Q^2\,q\sim^2 + 2\,Q^2 + 2 + 2\,q\sim^4 - 4\,q\sim^2,\ -2\,Q^2\,q\sim^2 + 2\,Q^2 - Q^4$$

```
> sigma1:=factor(mtaylor(e1[1],{q=1,Q},3));
  factor(solve(sigma1,{r}));
  sigma2:=factor(mtaylor(e1[2],{q=1,Q},4));
  solve(sigma2,{q});
```

$$\sigma 1 := -2\,r\sim + 8\,q\sim^2 - 16\,q\sim + 8$$

$$\{r\sim = 4\,(q\sim - 1)^2\}$$

$$\sigma 2 := -4\,Q^2\,(q\sim - 1)$$

$$\{q\sim = 1\} \qquad \textbf{(8)}$$

> # As we discussed in class, σ1 < σ2 (for r > 0), so ZZ stability boundary corresponds to σ2 = 0 (i.e., q = 1).

> # Eckhaus instability:

```
> eq:=collect(simplify(subs(a=sqrt(4/3*(r-(1-q^2)^2)),combine
  (coeff(N(a*cos(q*x)+epsilon*(alpha*cos(q*x)+beta*sin(q*x))*exp
  (I*Q*x)),epsilon,1),trig))),{cos(q*x),sin(q*x)}):
```

```
> A11:=simplify(coeff(coeff(eq*exp(-I*Q*x),cos(q*x),1),alpha,1)):
  A12:=simplify(coeff(coeff(eq*exp(-I*Q*x),cos(q*x),1),beta,1)):
  A21:=simplify(coeff(coeff(eq*exp(-I*Q*x),sin(q*x),1),alpha,1)):
  A22:=simplify(coeff(coeff(eq*exp(-I*Q*x),sin(q*x),1),beta,1)):
  A:=matrix(2,2,[A11,A12,A21,A22]);
  e1:=eigenvalues(A):
```

$$A := [[-6\,Q^2\,q\sim^2 + 2 - Q^4 - 4\,q\sim^2 + 2\,Q^2 + 2\,q\sim^4 - 2\,r\sim,\ 4\,\mathrm{I}\,Q\,q\sim\,(-1 + q\sim^2 + Q^2)],\ [-4\,\mathrm{I}\,Q\,q\sim\,(-1 + q\sim^2 + Q^2),\ -Q^2\,(6\,q\sim^2 + Q^2 - 2)]]$$

```
> sigma1:=factor(mtaylor(e1[1],{Q},3));
  sigma2:=mtaylor(e1[2],{q=1,Q},3);
```

$$\sigma 1 := -\frac{2\,Q^2\,(7\,q\sim^6 - 15\,q\sim^4 + 9\,q\sim^2 - 3\,r\sim\,q\sim^2 + r\sim - 1)}{-r\sim + 1 - 2\,q\sim^2 + q\sim^4}$$

$$\sigma 2 := -2\,r\sim + 8\,(q\sim - 1)^2 - 4\,Q^2$$

$$\sigma 2 := 2 + 2\,q\sim^4 - 4\,q\sim^2 - 2\,r\sim + \frac{2\,(q\sim^6 - q\sim^4 - q\sim^2 + 3\,r\sim\,q\sim^2 - r\sim + 1)}{-r\sim + 1 - 2\,q\sim^2 + q\sim^4}\,Q^2 + \mathrm{O}(Q^4)$$

> # Since σ2 < 0 for r > 0, stability boundary is determined by σ1 = 0:

```
> eqn:=coeff(sigma1,Q,2);
  rn(q):=solve(eqn,r);
  Rn(q):=series(subs(q=1+k,rn(q)),k,3);
```

$$eqn := -\frac{2\,(7\,q\sim^6 - 15\,q\sim^4 + 9\,q\sim^2 - 3\,r\sim\,q\sim^2 + r\sim - 1)}{-r\sim + 1 - 2\,q\sim^2 + q\sim^4}$$

Warning, solve may be ignoring assumptions on the input variables.

$$rn(q\sim) := \frac{7\,q\sim^6 - 15\,q\sim^4 + 9\,q\sim^2 - 1}{3\,q\sim^2 - 1}$$

$$Rn(q\sim) := 12\,k^2 + O(k^3)$$

```
> plot([rn(q),(1-q^2)^2],q=0.75..1.25);
```



green - neutral stability curve for uniform state
red - stability boundary for the stripe pattern